Object REXX for Windows News

Windows Scripting and GUI Extensions

Speaker: Christian Michel (cmichel@de.ibm.com)

REXX Development, IBM Germany
OLE/COM/ActiveX Class (1)

- Access scriptable applications from REXX:
  - MS Office products
  - Internet Explorer
  - Lotus Smartsuite products
  - Lotus Notes, etc.

- Automate activities in these applications:
  - Create and print customized letters
  - Update spreadsheets with statistics
  - Demo mode for Internet Explorer
OLE/COM/ActiveX Class (2)

- OLE objects can be created by their CLSID or ProgId
- All parameter types supported that can be converted to strings
- OLE arrays are converted into REXX .Array class objects in both directions
- Do..Over support for OLE collection types
OLE/COM/ActiveX Samples (1)

- Requires MS Common Dialogs Control
  (ships with MS development tools, such as Visual Studio, Visual C++, Visual Basic)

- Show File Open Dialog:
  ```
  ComDlg = .OLEObject~New("MSComDlg.CommonDialog")
  ComDlg~MaxFileSize = 256
  ComDlg~InitDir = Directory()
  ComDlg~Filter = "REXX Files|*.REX|REXX Class Definitions|*.CLS|Resource Scripts|*.RC"
  ComDlg~ShowOpen
  Say "Selected Filename:" ComDlg~FileName
  ::REQUIRES OREXXOLE.CLS
  ```

- Show Font Selection Dialog:
  ```
  ComDlg = .OLEObject~New("MSComDlg.CommonDialog")
  ComDlg~Flags = 1 /* cdlCFScreenFonts */
  ComDlg~ShowFont
  Say "Selected Font:" ComDlg~FontName || "." || ComDlg~FontSize
  ::REQUIRES OREXXOLE.CLS
  ```
Show Color Selection Dialog:

/* COLORDLG.REX - Use Common Dialogs */
ComDlg = .OLEObject~New("MSComDlg.CommonDialog")
ComDlg~Filter = 1 /* cdICCRGBInit */
ComDlg~ShowColor
RGB = ComDlg~Color~D2X~Right(6, "0")
Red = RGB~SubStr(5, 2)~X2D
Green = RGB~SubStr(3, 2)~X2D
Blue = RGB~SubStr(1, 2)~X2D
Say "Selected Color: Red =" Red "Green =" Green "Blue =" Blue

::REQUIRES OREXXOLE.CLS
OLE/COM/ActiveX Samples (3)

Display a list of URLs in sequence:

/* set the list of URLs to display */

/* create an object for Internet Explorer and tailor its appearance */
Explorer = .OLEObject~New("InternetExplorer.Application")
Explorer~TheaterMode = .True  /* show document area only */
Explorer~ToolBar = .False    /* don't show tool bar */
Explorer~Visible = .True     /* show Internet Explorer */

/* now loop through the URLs and display them */
Do Forever
  Do URL Over URLs
    Explorer~Navigate(URL)
    Call SysSleep 20
  End
End

::REQUIRES OREXXOLE.CLS
OLE/COM/ActiveX Samples (4)

Open Word with a document:

/* Create a new Word document and insert some text */
WordApp = .OLEObject~New("Word.Application")
WordDocs = WordApp~Documents
NewDocument = WordDocs~Add

/* Enter some text into first paragraph */
myRange = NewDocument~Paragraphs(1)~Range
myRange~Font~Name = "Brush Script"
myRange~Font~Size = "24"
myRange~InsertAfter("Hello from REXX!")
myRange~Words(3)~Bold = .True /* set the last word to bold */

/* Add a second paragraph with different attributes */
NewDocument~Paragraphs~Add
myRange = NewDocument~Paragraphs(2)~Range
myRange~Font~Name = "Arial"
myRange~Font~Size = "12"
myRange~Bold = .False
myRange~Italic = .False
myRange~InsertAfter("Greetings from the SHARE Technical Conference in SFO. ")
myRange~InsertAfter("This document was created on: ")
myRange~Collapse(0) /* unselect range, go to end (wdCollapseEnd=0) */
myRange~InsertDateTime("MMMM dd, yyyy at HH:mm:ss")
OLE/COM/ActiveX Samples (5)

Save and print a Word document:

/* continued from previous page */

/* Save and print document */
NewDocument~SaveAs("D:\Test.doc")
NewDocument~PrintOut
WordApp~Quit /* quit word */

::REQUIRES OREXXOLE.CLS
New OODIALOG Features

- **New Win32 Controls supported:**
  - List View (list, report, icon, small icon view)
  - Tree View
  - Tab View (notebook)
  - Progress Bar
  - Slider

- **Resource Workshop & Template Generator extended for new controls**
  - Workshop shows only outlines for new controls - still 16 bit application :-(
OODIALOG - List View (1)

- Four different views available:
  - List view (small icon and item name)
  - Report view (small icon and full item details)
  - Icon view (large icon and item name)
  - Small icon view (small icon and item name)
- Views can be switched on the fly
- Set icons for items in list (small & large)
- Insert/delete items in list
- Query all items in list
OODIALOG - List View (2)

- Insert/modify columns
- Get notifications for:
  - direct edit
  - activation (double click)
  - drag & drop
  - column click
  - key pressed
Activate notifications:

/* connect notifications in INIT method */
self~ConnectListNotify("IDC_LIST","Activate","OnActivate_IDC_LIST")

/* Method OnActivate_IDC_LIST handles notification 'Activate' for item IDC_LIST */
::METHOD OnActivate_IDC_LIST
    Expose curList
    info. = curList~ItemInfo(curList~Selected)
    If info.!Image = 0 Then Do
        /* change to the selected directory */
        curDir = self~GetValue("IDC_DIRECTORY")
        If curDir~Right(1) = \\
        Then curDir = curDir~Left(curDir~Length - 1)
        If info.!Text = ". . ." Then
            newDir = curDir~Left(curDir~LastPos("\\") - 1) || "\\" /* remove one directory from current path */
            Else
                newDir = curDir || "\\" || info.!Text || "\\"
        self~SetValue("IDC_DIRECTORY", newDir)
        self~IDC_REFRESH
    End
End
Setup report view and associated icons:

::METHOD InitDialog
   Expose curList
   InitDlgRet = self~InitDialog:super

   /* Set image list for List control IDC_LIST */
   curList = self~GetListControl("IDC_LIST")
   curList~SetSmallImages("ListSml.bmp",17,17)
   curList~SetImages("Listico.bmp",34,34)

   /* Add items (and columns) to the list control here using Add or AddRow (and InsertColumn) */
   curList~InsertColumn(0, "Filename", 50)
   curList~InsertColumn(1, "Size", 25, "RIGHT")
   curList~InsertColumn(2, "Date", 35)
   curList~InsertColumn(3, "Time", 30)
   curList~InsertColumn(4, "Attribs", 25)

   /* Initialization Code (e.g. fill list and combo boxes) */
   Return InitDlgRet
Delete and insert items:

::METHOD IDC_REFRESH /* Method IDC_REFRESH is connected to item IDC_REFRESH */
Expose curList
curList~DeleteAll
/* get a list of all files in specified directory */
curDir = self~GetValue("IDC_DIRECTORY")
If curDir~Right(1) \= "\" Then curDir = curDir || "\"
If curDir~Length > 3 Then curList~AddRow(0, "..")
pattern = curDir || "*"
Call SysFileTree pattern, "Files.", "BL"
Do i = 1 To Files.0
  FileName = Files.i~SubStr(Files.i~WordIndex(5))
  FileName = FileSpec("NAME", FileName)
  FileSize = Files.i~Word(3)
  FileDate = Files.i~Word(1)
  FileTime = Files.i~Word(2)
  FileAttribs = Files.i~Word(4)
  If FileAttribs~Pos("D") \= 0 Then FileIcon = 0
  Else FileIcon = 1
  curList~AddRow(FileIcon, FileName, FileSize, FileDate, FileTime, FileAttribs)
End
OODIALOG List View - Code

Samples (4)

Switch between views:

::METHOD ShowList /* Method ShowList is connected to item IDC_RADIODIALOG List */
   Expose curList curStyle
   curList~ReplaceStyle(curStyle, "LIST")
   curStyle = "LIST"

::METHOD ShowReport /* Method ShowReport is connected to item IDC_RADIODIALOG Report */
   Expose curList curStyle
   curList~ReplaceStyle(curStyle, "REPORT")
   curStyle = "REPORT"

::METHOD ShowIcon /* Method ShowIcon is connected to item IDC_RADIODIALOG ICON */
   Expose curList curStyle
   curList~ReplaceStyle(curStyle, "ICON")
   curStyle = "ICON"

::METHOD ShowSmallIcon /* Method ShowSmallIcon is connected to item IDC_RADIODIALOG Small ICON */
   Expose curList curStyle
   curList~ReplaceStyle(curStyle, "SMALL ICON")
   curStyle = "SMALL ICON"
OODIALOG List View - Sample Dialog

Object REXX List Sample

Directory: D:\Symp99\n
<table>
<thead>
<tr>
<th>Filename</th>
<th>Size</th>
<th>Date</th>
<th>Time</th>
<th>Attributes</th>
</tr>
</thead>
<tbody>
<tr>
<td>abstract.doc</td>
<td>1624</td>
<td>1999-03-22</td>
<td>16:33:18</td>
<td>A----</td>
</tr>
<tr>
<td>filedlg.rex</td>
<td>265</td>
<td>1999-04-28</td>
<td>14:42:20</td>
<td>A----</td>
</tr>
<tr>
<td>Pres</td>
<td>0</td>
<td>1999-04-29</td>
<td>18:35:42</td>
<td>AD----</td>
</tr>
<tr>
<td>WinNews</td>
<td>0</td>
<td>1999-04-29</td>
<td>18:35:54</td>
<td>AD----</td>
</tr>
<tr>
<td>WinUtil</td>
<td>0</td>
<td>1999-04-29</td>
<td>18:35:54</td>
<td>AD----</td>
</tr>
</tbody>
</table>

List: 
Report: 
Icon: 
Small Icon: 

OK
Cancel
Refresh
OODIALOG - Tree View

- Set icons for items in tree
- Expand/collapse full tree or parts of it
- Insert/delete/move/sort items
- Query all items in tree
- Get notifications for:
  - direct edit
  - drag & drop
  - selection
  - key pressed
Activate notifications:

/* connect notifications in INIT method */
self~ConnectTreeNotify("IDC_TREE","SelChanged","OnSelChanged_IDC_TREE")
self~ConnectTreeNotify("IDC_TREE","KeyDown","OnKeyDown_IDC_TREE")

/* Method OnSelChanged_IDC_TREE handles notification 'SelChanged' for item IDC_TREE */
::METHOD OnSelChanged_IDC_TREE
    Expose curTree selItem
    curTree~Modify(selItem,,,"NOTBOLD")
    selItem = curTree~Selected
    curTree~Modify(selItem,,,"BOLD")

/* Method OnKeyDown_IDC_TREE handles notification 'KeyDown' for item IDC_TREE */
::METHOD OnKeyDown_IDC_TREE
    Expose curTree selItem
    Use Arg treeld, Key

    /* if DELETE key is pressed delete the selected item */
    If self~KeyName(Key) = "DELETE" Then
        self~IDC_DELETE
Insert new items (a folder in this sample):

```rexx
/* Method IDC.NEWFOLDER is connected to item IDC.NEWFOLDER */
::METHOD IDC.NEWFOLDER
   Expose curTree selItem

   /* get new folder name */
   Input = .InputBox~New("Enter new folder name:", "Folder Name")
   newFldr = Input~Execute
   If newFldr != "" Then Do
      /* look at selected item */
      info. = curTree~ItemInfo(selItem)
      If info.!Image = 0 Then
         /* this is a folder, create new folder as a child */
         newPos = selItem
      Else
         /* this is an item, create new folder as a sibling */
         newPos = curTree~Parent(selItem)
         curTree~Insert(newPos, "LAST", newFldr, 0, 1)
         curTree~Expand(newPos)
   End
```

The 10th International Rexx Symposium, 3-5 May 1999, Jacksonville/Florida
List all items in tree

/* list items in tree, then terminate */
Say "The following items are in the tree:
Call ListItems curTree, curTree~Root, 0

::ROUTINE ListItems
  Use Arg treeCtrl, Item, Indent

  Do While Item \= 0
    info. = treeCtrl~ItemInfo(Item)
    Say Copies(" ", Indent) || info.!Text

    Child = treeCtrl~Child(Item)
    If Child \= 0 Then
      Call ListItems treeCtrl, Child, Indent + 2

    Item = treeCtrl~Next(Item)
  End
OODIALOG Tree View - Sample Dialog

![Object REXX Tree Control Sample](image)

Create your own tree structure:
- root
  - Folder 1
    - Item 1
    - Item 2
  - Folder 2
  - My Hobbies
    - Ballooning
    - Flying
    - Cycling
    - Computers
OODIALOG - Progress Bar/Slider

- Customizable progress indicator
  - horizontal/vertical
  - smooth/discrete progress
  - minimum and maximum values

- Customizable slider input control
  - horizontal/vertical
  - tick mark locations/intervals
  - notifications for user changes
OODIALOG - Progress Bar & Slider Samples

- File copy progress dialog:

![File copy progress dialog](image1)

- Slider sample:

![Slider sample](image2)
Further Information

**Object REXX Homepage:**
http://www.software.ibm.com/ad/obj-rexx/

**Object REXX Service & Support Page:**
http://service.software.ibm.com/dl/rexx/orexx-d